NAG Library Routine Document

F12FBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

Note: *this routine uses* **optional parameters** *to define choices in the problem specification. If you wish to use* default *settings for all of the optional parameters, then the option setting routine F12FDF need not be called. If, however, you wish to reset some or all of the settings please refer to Section 11 in F12FDF for a detailed description of the specification of the optional parameters*.

1 Purpose

F12FBF is an iterative solver in a suite of routines consisting of F12FAF, F12FBF, F12FCF, F12FDF and F12FEF. It is used to find some of the eigenvalues (and optionally the corresponding eigenvectors) of a standard or generalized eigenvalue problem defined by real symmetric matrices.

2 Specification

```
SUBROUTINE F12FBF (IREVCM, RESID, V, LDV, X, MX, NSHIFT, COMM, ICOMM,      &
                   IFAIL)
INTEGER            IREVCM, LDV, NSHIFT, ICOMM(*), IFAIL
REAL (KIND=nag_wp) RESID(*), V(LDV,*), X(*), MX(*), COMM(*)
```

3 Description

The suite of routines is designed to calculate some of the eigenvalues, λ, (and optionally the corresponding eigenvectors, x) of a standard eigenvalue problem $Ax = \lambda x$, or of a generalized eigenvalue problem $Ax = \lambda Bx$ of order n, where n is large and the coefficient matrices A and B are sparse, real and symmetric. The suite can also be used to find selected eigenvalues/eigenvectors of smaller scale dense, real and symmetric problems.

F12FBF is a **reverse communication** routine, based on the ARPACK routine **dsaupd**, using the Implicitly Restarted Arnoldi iteration method, which for symmetric problems reduces to a variant of the Lanczos method. The method is described in Lehoucq and Sorensen (1996) and Lehoucq (2001) while its use within the ARPACK software is described in great detail in Lehoucq *et al.* (1998). An evaluation of software for computing eigenvalues of sparse symmetric matrices is provided in Lehoucq and Scott (1996). This suite of routines offers the same functionality as the ARPACK software for real symmetric problems, but the interface design is quite different in order to make the option setting clearer and to simplify the interface of F12FBF.

The setup routine F12FAF must be called before F12FBF, the reverse communication iterative solver. Options may be set for F12FBF by prior calls to the option setting routine F12FDF and a post-processing routine F12FCF must be called following a successful final exit from F12FBF. F12FEF, may be called following certain flagged, intermediate exits from F12FBF to provide additional monitoring information about the computation.

F12FBF uses **reverse communication**, i.e., it returns repeatedly to the calling program with the parameter IREVCM (see Section 5) set to specified values which require the calling program to carry out one of the following tasks:

– compute the matrix-vector product $y = \mathrm{OP}x$, where OP is defined by the computational mode;

– compute the matrix-vector product $y = Bx$;

– notify the completion of the computation;

– allow the calling program to monitor the solution.

The problem type to be solved (standard or generalized), the spectrum of eigenvalues of interest, the mode used (regular, regular inverse, shifted inverse, Buckling or Cayley) and other options can all be set using the option setting routine F12FDF.

4 References

Lehoucq R B (2001) Implicitly restarted Arnoldi methods and subspace iteration *SIAM Journal on Matrix Analysis and Applications* **23** 551–562

Lehoucq R B and Scott J A (1996) An evaluation of software for computing eigenvalues of sparse nonsymmetric matrices *Preprint MCS-P547-1195* Argonne National Laboratory

Lehoucq R B and Sorensen D C (1996) Deflation techniques for an implicitly restarted Arnoldi iteration *SIAM Journal on Matrix Analysis and Applications* **17** 789–821

Lehoucq R B, Sorensen D C and Yang C (1998) *ARPACK Users' Guide: Solution of Large-scale Eigenvalue Problems with Implicitly Restarted Arnoldi Methods* SIAM, Philidelphia

5 Parameters

Note: this routine uses **reverse communication.** Its use involves an initial entry, intermediate exits and re-entries, and a final exit, as indicated by the parameter **IREVCM**. Between intermediate exits and re-entries, **all parameters other than X, MX and COMM must remain unchanged**.

1: IREVCM – INTEGER *Input/Output*

On initial entry: IREVCM $= 0$, otherwise an error condition will be raised.

On intermediate re-entry: must be unchanged from its previous exit value. Changing IREVCM to any other value between calls will result in an error.

On intermediate exit: has the following meanings.

IREVCM $= -1$
The calling program must compute the matrix-vector product $y = \mathrm{OP}x$, where x is stored in X (by default) or in the array COMM (starting from the location given by the first element of ICOMM) when the option **Pointers** = YES is set in a prior call to F12FDF. The result y is returned in X (by default) or in the array COMM (starting from the location given by the second element of ICOMM) when the option **Pointers** = YES is set.

IREVCM $= 1$
The calling program must compute the matrix-vector product $y = \mathrm{OP}x$. This is similar to the case IREVCM $= -1$ except that the result of the matrix-vector product Bx (as required in some computational modes) has already been computed and is available in MX (by default) or in the array COMM (starting from the location given by the third element of ICOMM) when the option **Pointers** = YES is set.

IREVCM $= 2$
The calling program must compute the matrix-vector product $y = Bx$, where x is stored in X and y is returned in MX (by default) or in the array COMM (starting from the location given by the second element of ICOMM) when the option **Pointers** = YES is set.

IREVCM $= 3$
Compute the NSHIFT real and imaginary parts of the shifts where the real parts are to be returned in the first NSHIFT locations of the array X and the imaginary parts are to be returned in the first NSHIFT locations of the array MX. Only complex conjugate pairs of shifts may be applied and the pairs must be placed in consecutive locations. This value of IREVCM will only arise if the optional parameter **Supplied Shifts** is set in a prior call to F12FDF which is intended for experienced users only; the default and recommended option is to use exact shifts (see Lehoucq *et al.* (1998) for details and guidance on the choice of shift strategies).

IREVCM = 4
Monitoring step: a call to F12FEF can now be made to return the number of Arnoldi iterations, the number of converged Ritz values, their real and imaginary parts, and the corresponding Ritz estimates.

On final exit: IREVCM = 5: F12FBF has completed its tasks. The value of IFAIL determines whether the iteration has been successfully completed, or whether errors have been detected. On successful completion F12FCF must be called to return the requested eigenvalues and eigenvectors (and/or Schur vectors).

Constraint: on initial entry, IREVCM = 0; on re-entry IREVCM must remain unchanged.

2: RESID(∗) – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array RESID must be at least N (see F12FAF).

On initial entry: need not be set unless the option **Initial Residual** has been set in a prior call to F12FDF in which case RESID should contain an initial residual vector, possibly from a previous run.

On intermediate re-entry: must be unchanged from its previous exit. Changing RESID to any other value between calls may result in an error exit.

On intermediate exit: contains the current residual vector.

On final exit: contains the final residual vector.

3: V(LDV, ∗) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array V must be at least max(1, NCV) (see F12FAF).

On initial entry: need not be set.

On intermediate re-entry: must be unchanged from its previous exit.

On intermediate exit: contains the current set of Arnoldi basis vectors.

On final exit: contains the final set of Arnoldi basis vectors.

4: LDV – INTEGER *Input*

On entry: the first dimension of the array V as declared in the (sub)program from which F12FBF is called.

Constraint: LDV ≥ N.

5: X(∗) – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array X must be at least N if **Pointers** = NO (default) and at least 1 if **Pointers** = YES (see F12FAF).

On initial entry: need not be set, it is used as a convenient mechanism for accessing elements of COMM.

On intermediate re-entry: if **Pointers** = YES, X need not be set.

If **Pointers** = NO, X must contain the result of $y = \mathrm{OP}x$ when IREVCM returns the value -1 or $+1$. It must return the real parts of the computed shifts when IREVCM returns the value 3.

On intermediate exit: if **Pointers** = YES, X is not referenced.

If **Pointers** = NO, X contains the vector x when IREVCM returns the value -1 or $+1$.

On final exit: does not contain useful data.

6: MX($*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array MX must be at least N if **Pointers** = NO (default) and at least 1 if **Pointers** = YES (see F12FAF).

On initial entry: need not be set, it is used as a convenient mechanism for accessing elements of COMM.

On intermediate re-entry: if **Pointers** = YES, MX need not be set.

If **Pointers** = NO, MX must contain the result of $y = Bx$ when IREVCM returns the value 2. It must return the imaginary parts of the computed shifts when IREVCM returns the value 3.

On intermediate exit: if **Pointers** = YES, MX is not referenced.

If **Pointers** = NO, MX contains the vector Bx when IREVCM returns the value $+1$.

On final exit: does not contain any useful data.

7: NSHIFT – INTEGER *Output*

On intermediate exit: if the option **Supplied Shifts** is set and IREVCM returns a value of 3, NSHIFT returns the number of complex shifts required.

8: COMM($*$) – REAL (KIND=nag_wp) array *Communication Array*

Note: the dimension of the array COMM must be at least $\max(1, \text{LCOMM})$ (see F12FAF).

On initial entry: must remain unchanged following a call to the setup routine F12FAF.

On exit: contains data defining the current state of the iterative process.

9: ICOMM($*$) – INTEGER array *Communication Array*

Note: the dimension of the array ICOMM must be at least $\max(1, \text{LICOMM})$ (see F12FAF).

On initial entry: must remain unchanged following a call to the setup routine F12FAF.

On exit: contains data defining the current state of the iterative process.

10: IFAIL – INTEGER *Input/Output*

On initial entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On final exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On initial entry, the maximum number of iterations ≤ 0, the option **Iteration Limit** has been set to a non-positive value.

IFAIL $= 2$

The options **Generalized** and **Regular** are incompatible.

IFAIL $= 3$

Eigenvalues from both ends of the spectrum were requested, but the number of eigenvalues requested is one.

IFAIL $= 4$

The option **Initial Residual** was selected but the starting vector held in RESID is zero.

IFAIL $= 5$

The maximum number of iterations has been reached. Some Ritz values may have converged; a subsequent call to F12FCF will return the number of converged values and the converged values.

IFAIL $= 6$

No shifts could be applied during a cycle of the implicitly restarted Arnoldi iteration. One possibility is to increase the size of NCV relative to NEV (see Section 5 in F12FAF for details of these parameters).

IFAIL $= 7$

Could not build a Lanczos factorization. Consider changing NCV or NEV in the initialization routine (see Section 5 in F12FAF for details of these parameters).

IFAIL $= 8$

Unexpected error in internal call to compute eigenvalues and corresponding error bounds of the current upper Hessenberg matrix. Please contact NAG.

IFAIL $= 9$

An unexpected error has occurred. Please contact NAG.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

The relative accuracy of a Ritz value, λ, is considered acceptable if its Ritz estimate $\leq$ **Tolerance** $\times |\lambda|$. The default **Tolerance** used is the ***machine precision*** given by X02AJF.

8 Parallelism and Performance

F12FBF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F12FBF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

For this routine two examples are presented, with a main program and two example problems given in Example 1 (EX1) and Example 2 (EX2).

Example 1 (EX1)

The example solves $Ax = \lambda x$ in shift-invert mode, where A is obtained from the standard central difference discretization of the one-dimensional Laplacian operator $\frac{\partial^2 u}{\partial x^2}$ with zero Dirichlet boundary conditions. Eigenvalues closest to the shift $\sigma = 0$ are sought.

Example 2 (EX2)

This example illustrates the use of F12FBF to compute the leading terms in the singular value decomposition of a real general matrix A. The example finds a few of the largest singular values (σ) and corresponding right singular values (ν) for the matrix A by solving the symmetric problem:

$$\left(A^{\mathrm{T}}A\right)\nu = \sigma\nu.$$

Here A is the m by n real matrix derived from the simplest finite difference discretization of the two-dimensional kernal $k(s,t)dt$ where

$$k(s,t) = \begin{cases} s(t-1) & \text{if } 0 \leq s \leq t \leq 1 \\ t(s-1) & \text{if } 0 \leq t < s \leq 1 \end{cases}.$$

Note: this formulation is appropriate for the case $m \geq n$. Reverse the rules of A and A^{T} in the case of $m < n$.

10.1 Program Text

```
!   F12FBF Example Program Text
!   Mark 25 Release. NAG Copyright 2014.

    Module f12fbfe_mod

!     F12FBF Example Program Module:
!            Parameters and User-defined Routines

!     .. Use Statements ..
      Use nag_library, Only: nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Accessibility Statements ..
      Private
      Public                                  :: atv, av
!     .. Parameters ..
      Real (Kind=nag_wp), Parameter, Public :: one = 1.0_nag_wp
      Real (Kind=nag_wp), Parameter, Public :: two = 2.0_nag_wp
      Real (Kind=nag_wp), Parameter, Public :: zero = 0.0_nag_wp
      Integer, Parameter, Public            :: imon = 0, licomm = 140, nin = 5, &
                                               nout = 6
    Contains
      Subroutine av(m,n,x,w)
```

```
!       Computes  w <- A*x.

!       .. Scalar Arguments ..
        Integer, Intent (In)                 :: m, n
!       .. Array Arguments ..
        Real (Kind=nag_wp), Intent (Out)     :: w(m)
        Real (Kind=nag_wp), Intent (In)      :: x(n)
!       .. Local Scalars ..
        Real (Kind=nag_wp)                   :: h, k, s, t
        Integer                              :: i, j
!       .. Intrinsic Procedures ..
        Intrinsic                            :: real
!       .. Executable Statements ..
        h = one/real(m+1,kind=nag_wp)
        k = one/real(n+1,kind=nag_wp)
        w(1:m) = zero
        t = zero

        Do j = 1, n
          t = t + k
          s = zero
          Do i = 1, j
            s = s + h
            w(i) = w(i) + k*s*(t-one)*x(j)
          End Do
          Do i = j + 1, m
            s = s + h
            w(i) = w(i) + k*t*(s-one)*x(j)
          End Do
        End Do

        Return
      End Subroutine av

      Subroutine atv(m,n,w,y)

!       Computes  y <- A'*w.

!       .. Scalar Arguments ..
        Integer, Intent (In)                 :: m, n
!       .. Array Arguments ..
        Real (Kind=nag_wp), Intent (In)      :: w(m)
        Real (Kind=nag_wp), Intent (Out)     :: y(n)
!       .. Local Scalars ..
        Real (Kind=nag_wp)                   :: h, k, s, t
        Integer                              :: i, j
!       .. Intrinsic Procedures ..
        Intrinsic                            :: real
!       .. Executable Statements ..
        h = one/real(m+1,kind=nag_wp)
        k = one/real(n+1,kind=nag_wp)
        y(1:n) = zero
        t = zero

        Do j = 1, n
          t = t + k
          s = zero
          Do i = 1, j
            s = s + h
            y(j) = y(j) + k*s*(t-one)*w(i)
          End Do
          Do i = j + 1, m
            s = s + h
            y(j) = y(j) + k*t*(s-one)*w(i)
          End Do
        End Do

        Return
      End Subroutine atv
    End Module f12fbfe_mod
    Program f12fbfe
```

```
!     F12FBF Example Main Program

!     .. Use Statements ..
      Use f12fbfe_mod, Only: nout
!     .. Implicit None Statement ..
      Implicit None
!     .. Executable Statements ..
      Write (nout,*) 'F12FBF Example Program Results'

      Call ex1

      Call ex2

    Contains
      Subroutine ex1

!       .. Use Statements ..
        Use nag_library, Only: dgttrf, dgttrs, dnrm2, f12faf, f12fbf, f12fcf,  &
                               f12fdf, f12fef, nag_wp
        Use f12fbfe_mod, Only: imon, licomm, nin, one, two, zero
!       .. Local Scalars ..
        Real (Kind=nag_wp)                   :: h2, sigma
        Integer                              :: ifail, info, irevcm, j, lcomm, &
                                                ldv, n, nconv, ncv, nev,       &
                                                niter, nshift
!       .. Local Arrays ..
        Real (Kind=nag_wp), Allocatable      :: ad(:), adl(:), adu(:),         &
                                                adu2(:), comm(:), d(:,:),      &
                                                mx(:), resid(:), v(:,:), x(:)
        Integer                              :: icomm(licomm)
        Integer, Allocatable                 :: ipiv(:)
!       .. Intrinsic Procedures ..
        Intrinsic                            :: real
!       .. Executable Statements ..
        Write (nout,*)
        Write (nout,*) 'F12FBF Example 1'
        Write (nout,*)
        Flush (nout)
!       Skip heading in data file
        Read (nin,*)
        Read (nin,*)
        Read (nin,*) n, nev, ncv

        ldv = n
        lcomm = 3*n + ncv*ncv + 8*ncv + 60
        Allocate (ad(n),adl(n),adu(n),adu2(n),comm(lcomm),d(ncv,2),mx(n), &
          resid(n),v(ldv,ncv),x(n),ipiv(n))

        ifail = 0
        Call f12faf(n,nev,ncv,icomm,licomm,comm,lcomm,ifail)

!       Set the region of the spectrum that is required.
        Call f12fdf('LARGEST MAGNITUDE',icomm,comm,ifail)
!       Use the Shifted Inverse mode.
        Call f12fdf('SHIFTED INVERSE',icomm,comm,ifail)

        h2 = one/real((n+1)*(n+1),kind=nag_wp)
        sigma = zero
        ad(1:n) = two/h2 - sigma
        adl(1:n) = -one/h2
        adu(1:n) = adl(1:n)
!       The NAG name equivalent of dgttrf is f07cdf
        Call dgttrf(n,adl,ad,adu,adu2,ipiv,info)

        irevcm = 0
        ifail = -1
revcm:  Do
          Call f12fbf(irevcm,resid,v,ldv,x,mx,nshift,comm,icomm,ifail)
          If (irevcm==5) Then
            Exit revcm
```

```
          Else If (irevcm==-1 .Or. irevcm==1) Then
!           Perform matrix vector multiplication
!           y <--- inv[A-sigma*I]*x
!           The NAG name equivalent of dgttrs is f07cef
            Call dgttrs('N',n,1,adl,ad,adu,adu2,ipiv,x,n,info)
          Else If (irevcm==4 .And. imon/=0) Then
!           Output monitoring information
            Call f12fef(niter,nconv,d,d(1,2),icomm,comm)
!           The NAG name equivalent of dnrm2 is f06ejf
            Write (6,99999) niter, nconv, dnrm2(nev,d(1,2),1)
            Flush (nout)
          End If
        End Do revcm

        If (ifail==0) Then
!         Post-Process using F12FCF to compute eigenvalues/vectors.
          Call f12fcf(nconv,d,v,ldv,sigma,resid,v,ldv,comm,icomm,ifail)
          Write (nout,99998) nconv, sigma
          Write (nout,99997)(j,d(j,1),j=1,nconv)
        End If

        Deallocate (ad,adl,adu,adu2,comm,d,mx,resid,v,x,ipiv)

        Return

99999   Format (1X,'Iteration',1X,I3,', No. converged =',1X,I3,', norm o', &
          'f estimates =',E16.8)
99998   Format (1X/' The ',I4,' Ritz values of closest to ',F8.4,' are:'/)
99997   Format (1X,I8,5X,F12.4)
      End Subroutine ex1
      Subroutine ex2

!       .. Use Statements ..
        Use nag_library, Only: daxpy, dnrm2, dscal, f12faf, f12fbf, f12fcf,    &
                               nag_wp, x04caf
        Use f12fbfe_mod, Only: atv, av, licomm, nin, one
!       .. Local Scalars ..
        Real (Kind=nag_wp)                   :: sigma, temp
        Integer                              :: ifail, ifail1, irevcm, j,      &
                                                lcomm, ldu, ldv, m, n, nconv,  &
                                                ncv, nev, nshift
!       .. Local Arrays ..
        Real (Kind=nag_wp), Allocatable      :: ax(:), comm(:), d(:,:), mx(:), &
                                                resid(:), u(:,:), v(:,:), x(:)
        Integer                              :: icomm(licomm)
!       .. Intrinsic Procedures ..
        Intrinsic                            :: sqrt
!       .. Executable Statements ..
        Write (nout,*)
        Write (nout,*) 'F12FBF Example 2'
        Write (nout,*)
        Flush (nout)
!       Skip heading in data file
        Read (nin,*)
        Read (nin,*) m, n, nev, ncv

        ldu = m
        ldv = n
        lcomm = 3*n + ncv*ncv + 8*ncv + 60
        Allocate (ax(m),comm(lcomm),d(ncv,2),mx(m),resid(n),u(ldu,nev), &
          v(ldv,ncv),x(m))

!       Initialize for problem.
        ifail = 0
        Call f12faf(n,nev,ncv,icomm,licomm,comm,lcomm,ifail)

!       Main reverse communication loop.
        irevcm = 0
        ifail = -1
revcm:  Do
          Call f12fbf(irevcm,resid,v,ldv,x,mx,nshift,comm,icomm,ifail)
```

```
          If (irevcm==5) Then
            Exit revcm
          Else If (irevcm==-1 .Or. irevcm==1) Then
!           Perform the operation X <- A'AX
            Call av(m,n,x,ax)
            Call atv(m,n,ax,x)
          End If
        End Do revcm

        If (ifail==0) Then
!         Post-Process using F12FCF.
!         Computed singular values may be extracted.
!         Singular vectors may also be computed now if desired.
          ifail1 = 0
          Call f12fcf(nconv,d,v,ldv,sigma,resid,v,ldv,comm,icomm,ifail1)

!         Singular values (squared) are returned in the first column
!         of D and the corresponding right singular vectors are
!         returned in the first NEV columns of V.

          Do j = 1, nconv
            d(j,1) = sqrt(d(j,1))

!           Compute the left singular vectors from the formula
!           u = Av/sigma/norm(Av).

            Call av(m,n,v(1,j),ax)
            u(1:m,j) = ax(1:m)
!           The NAG name equivalent of dnrm2 is f06ejf
            temp = one/dnrm2(m,u(1,j),1)
!           The NAG name equivalent of dscal is f06edf
            Call dscal(m,temp,u(1,j),1)

!           Compute the residual norm ||A*v - sigma*u|| for the nconv
!           accurately computed singular values and vectors.
!           Store the result in 2nd column of array D.

!           The NAG name equivalent of daxpy is f06ecf
            Call daxpy(m,-d(j,1),u(1,j),1,ax,1)
            d(j,2) = dnrm2(m,ax,1)

          End Do

!         Print computed residuals

          Call x04caf('G','N',nconv,2,d,ncv, &
            ' Singular values and direct residuals',ifail1)
        End If

        Deallocate (ax,comm,d,mx,resid,u,v,x)

        Return

      End Subroutine ex2
    End Program f12fbfe
```

10.2 Program Data

```
F12FBF Example Program Data
Example 1
 100 4 10 : Values for N NEV and NCV
Example 2
 500 100 4 10 : Values for M N NEV and NCV
```

10.3 Program Results

```
 F12FBF Example Program Results

 F12FBF Example 1


 The    4 Ritz values of closest to   0.0000 are:

        1           9.8688
        2          39.4657
        3          88.7620
        4         157.7101

 F12FBF Example 2

  Singular values and direct residuals
                1            2
1    4.1012E-02   1.1877E-17
2    6.0488E-02   2.6231E-17
3    1.1784E-01   1.6455E-17
4    5.5723E-01   3.2163E-16
```